APPENDIX C

COMPUTER PROGRAM AXILTR

C-1. **Organization**

Program AXILTR, AXIal Load-TRansfeR, consists of a main routine and two subroutines. The main routine feeds in the input data, calculates the effective overburden stress, and determines whether the load is axial down-directed, pullout, or if uplift/downdrag forces develop from swelling or consolidating soil. The main routine also prints out the computations. Subroutine BASEL calculates the displacement at the base for given applied down-directed loads at the base. Subroutine SHAFL evaluates the load transferred to and from the shaft for relative displacements between the shaft and soil. An iteration scheme is used to cause the calculated applied loads at the top (butt) to converge within 10 percent of the input load applied at the top of the shaft.

a. **Input Data**

Input data are illustrated in Table C-1 with descriptions given in Table C-2.

TABLE C-1

Input Data

Line	Input Parameters	Format Statement
1	TITLE	20A4
2	NMAT NEL DX GWL LO IQ IJ	2I5,2F6.2,3I5
3	I J K SOILP DS DB	3I5,3F10.3
4	E50 (Omitted unless K = 2, 5, 9)	E13.3
5	LLL	I5
6	MAT GS EO WO PS CS CC C PHI AK PM (Line 5 repeated for each material M = 1,NMAT)	I3,3F6.2,F7.0, 2F7.2,F7.0,2F6.2 F7.0
7	ALPHA (Omitted unless I = 6) (α input for each material MAT = 1,NMAT)	7F10.5
8	M IE(M) (Line 8 repeated for each element M and number of soil IE(M). Start with 1. The last line is NEL NMAT)	2I5
9	RFF GG (Omitted unless K = 7, 8, 9)	F6.3,E13.3
10	(Omitted unless K = 3,4,5,6)	
10a	NCA (<12)	I5

TABLE C-1 (Concluded)

Line	Input Parameters	Format Statement
10b	T(M,1)... T(M,11) (Input for each curve M=1,NCA)	11F6.2
10c	S(M) (Input on new line for each M = 2,11; S(1) input in program as 0.00)	F6.3
11	(Omitted unless I = 5)	
11a	NCC (<12)	I5
11b	FS(N) ZEPP(N) NCUR (Input on new line for each N = 1,NCC)	2F10.3,I5
12	(Omitted unless J = 0)	
12a	NC (>1)	I5
12b	EP(M) ZEP(M) (Input on new line for each M = 1,NC; at least a top and bottom term required)	E13.3,F6.2
13	R(M) S(M) (Omitted unless K = 6; repeat on new line for each M = 1,IJ)	F10.5,F15.3
14	STRUL SOILP XA	3F15.2
15	NON (Omitted unless XA < 0.0)	I5

TABLE C-2

Description of Input Parameters

Line	Parameter	Description
1	TITLE	Name of problem
2	NMAT	Total number of materials
	NEL	Total number of elements
	DX	Thickness of each element, ft (usually 0.5 or 1.0 ft)
	GWL	Depth to groundwater level, ft
	LO	Amount of output data = 0 Extensive data output used to check the program = 1 Shaft load-displacement behavior and detailed load distribution-displacement response along shaft length for input top load prior to and following soil movement (load transfer, load, shaft compression increment, and shaft movement at given depth

TABLE C-2 (Continued)

Line	Parameter	Description
		= 2 Shaft load-displacement behavior and load distribution-displacement response along shaft length for input top load prior to and following soil movement
		= 3 Shaft load-displacement behavior and load distribution-displacement response along shaft length for input top load on shaft following soil movement
	IQ	Total number of shaft increments (shaft length/element thickness); top of shaft at ground surface
	IJ	Number of points for shaft load-displacement behavior (usually 12, but maximum of 18 for PARAMETER statement = 40)
3	I	Magnitude of reduction factor α applied to total (undrained) or effective (drained) shear strength for skin friction resistance = 0 $\alpha = 1$ (usually used for drained strength) = 1 $\alpha = \sin(\pi x/L)$, x = depth, ft; L = shaft length, ft = 2 $\alpha = 0.6$ = 3 $\alpha = 0.45$ = 4 $\alpha = 0.3$ = 5 Permits maximum skin friction f_s input as a function of depth, psf (see line 11) = 6 α is input for each material (see line 7)
	J	Option for elastic shaft modulus = 0 shaft modulus input = 1 shaft modulus set to near infinity
	K	Option for load-transfer functions (see Figure 5-12)

	Base	Shaft
= 0	Consolidation	Seed and Reese
= 1	Vijayvergiya	Seed and Reese
= 2	Reese and Wright	Seed and Reese
= 3	Consolidation	Input (see line 10)
= 4	Vijayvergiya	Input (see line 10)
= 5	Reese and Wright	Input (see line 10)
= 6	Input (see line 13)	Input (see line 10)
= 7	Consolidation	Kraft, Ray, and Kagawa
= 8	Vijayvergiya	Kraft, Ray, and Kagawa
= 9	Reese and Wright	Kraft, Ray, and Kagawa

TABLE C-2 (Continued)

Line	Parameter	Description
	SOILP	Pressure on top layer of soil exerted by surrounding structure, fill, etc., psf
	DS	Diameter shaft, ft
	DB	Diameter base, ft
4	E50	Strain at 1/2 maximum deviator stress, Equation 5-19
5	LLL	Option for type of shear failure at base = 0 Local shear failure, Equation 5-7 or N_c = 7 = 1 General shear failure, Equation 5-8 or N_c = 9
6	MAT	Number of material
	GS	Specific gravity
	EO	Initial void ratio
	WO	Initial water content, percent
	PS	Swell pressure, psf
	CS	Swell index
	CC	Compression index
	C	Cohesion, psf; = undrained strength for total stress analysis; effective cohesion c′ or zero for effective stress analysis
	PHI	Angle of shearing resistance ϕ; = 0 for total stress analysis
	AK	Coefficient of lateral earth pressure
	PM	Maximum past pressure, psf (program sets PM = PS if PM input < PS)
7	ALPHA	Reduction factor α_a for each material MAT, Equation 5-11, Table 5-1, Table 5-10; used when option I = 6, Line 3
8	M	Number of element
	IE(M)	Material number of soil, MAT
9	RFF	Hyperbolic reduction factor R for Kraft, Ray, and Kagawa model, Equation 5-19a; use 1.0 if not known
	GG	Shear modulus G_s, psf, Equation 5-19b
10		Input data for shaft load-transfer curves (K = 3,4,5,6)
10a	NCA	Total number of shaft load-transfer curves to input, < 12
10b	T(M,1)... ..T(M,11)	Skin friction ratio of developed shear strength/maximum mobilized shear strength of each shaft load-transfer curve; 11 values required for each load-transfer curve, the first value T(1,1) = 0.0

TABLE C-2 (Concluded)

Line	Parameter	Description
10c	S(M)	Movement in inches for all of the T(M,1)...T(M,11) curves; only 10 values required from S(2)...S(11); S(1) = 0.0 in code; if S(M) in the code is okay (0.0,0.05,0.1,0.15,0.2, 0.23,0.3,0.45,0.75,1.05,1.5 inches)
11		Input data for maximum skin friction as a function of depth
	NCC	Total number of maximum skin friction terms to input, <12; program interpolates maximum skin friction between depths
11a	FS(N)	Maximum skin friction f_s for point N, psf
11b	ZEPP(N)	Depth for the maximum skin friction for point N, ft
11c	NCUR	Number of the shaft load-transfer curve input M in line 10; applicable to the maximum skin friction for point N (Repeat 11a,11b,11c for each N = 1,NCC)
12		Input data for shaft elastic modulus as function of depth; program interpolates the elastic modulus between depths
	NC	Total number of terms of elastic modulus and depth, > 1
12a	EP(M)	Elastic modulus of shaft at point M, psf
12b	ZEP(M)	Depth for the elastic modulus of shaft at point M, ft (An elastic modulus and depth term are required at least at the top and bottom of the shaft)
13		Input data for base displacements if K = 6 (The number of input terms or R(M) and S(M) equals IJ - 1, line 2)
13a	R(M)	Base displacement, in. (The first displacement is 0.0 inches and already input in the program)
13b	S(M)	Base load for displacement R(M), pounds; the base load for 0.0 displacement is approximated as the overlying soil weight and already input in the program.
14		Structural load, pressure on adjacent soil at the ground surface, and depth of the active zone for heave input for each problem for evaluation of specific load distribution-displacement computations
14a	STRUC	Structural vertical load on top of shaft, pounds
14b	SOILP	Pressure on top layer of soil exerted by surrounding structure, fill, etc., psf
14c	XA	Depth of the active zone for heave, ft; = 0.01 yields load-displacement behavior for zero soil movement; a saturated soil profile is assumed when computing soil movement; < 0.0 program goes to line 15 below
15	NON	Execution stops if 0; program goes to line 1 if > 0

(1) The program is set to consider up to a total of 40 soil types and 100 soil elements. Figure C-1 provides an example layout of soil types and elements used in AXILTR.

(2) The program can accommodate up to 18 points of the load-displacement curve. This capacity may be altered by adjusting the PARAMETER statement in the program.

(3) The input data are placed in a file, "DATLTR.TXT". These data are printed in output file, "LTROUT.TXT" illustrated in Table C-3a.



Figure C-1. Schematic diagram of soil/pile elements

TABLE C-3

Output Data

a. Repeat of Input Data (See Table C-1)

Line	Output Parameters	Fortran Statement
1	TITLE	20A4
2	NMAT= NEL= DX= FT GWL= FT LO= IQ (SHAFT INC)= IJ (NO.LOADS)=	I5,I5,F6.2,F6.2 I5,I5,I5
3	I= J= K= SOILP= PSF DS= FT DB= FT	I5,I5,I5,F10.2 F10.2 F10.2
4	(If K = 2,5,9) E50	 E13.3
5	LOCAL SHEAR FAILURE AT BASE - LLL = 0 or GENERAL SHEAR FAILURE AT BASE - LLL = 1	I5 I5
6	MAT GS EO WO(%) PS(PSF) CS CC CO(PSF) PHI K PM(PSF)	I3,3F6.2,F7.0, 27.2,F7.0,2F6.2, F7.0
7	(If I = 6) ALPHA =	2(7F10.5)
8	ELEMENT NO OF SOIL	I5,10X,I5
9	(If K = 7,8,9) REDUCTION FACTOR= SHEAR MODULUS=	 F6.3,3X,E13.3
10	(If K = 3,4,5,6) NO. OF LOAD-TRANSFER CURVES(<12)?=	 I5
	For each curve 1 to NCA: CURVE	 I5
	RATIO SHR DEV,M=1,11 ARE MOVEMENT(IN.) FOR LOAD TRANSFER M= IS INCHES	11F6.3 I5,F6.3
11	(If I = 5) NO OF SKIN FRICTION-DEPTH TERMS(<12)? ARE SKIN FRICTION(PSF) DEPTH(FT) CURVE NO	 I5 F10.3,F10.3,I5
12	(If J = 0) E SHAFT(PSF) AND DEPTH(FT):	 4(E13.3,2X,F6.2)
13	(If K = 6) BASE DISPLACEMENT(IN.),BASE LOAD(LBS) > FOR POINTS	 F10.2,I5

TABLE C-3 (Continued)

b. Output Calculations

Item	Program Prints	Format Statement
1	BEARING CAPACITY= POUNDS	F13.2
2	DOWNWARD DISPLACEMENT	
3	(Omitted unless LO = 0,1) POINT BEARING(LBS)=	F13.2
4	(Omitted unless LO = 0,1) DEPTH LOAD TRANS TOTAL LOAD COM OF INCR TOTAL MOVMT ITER FT LBS LBS INCHES INCHES	5E13.5,I5
5	TOP LOAD TOP MOVEMENT BASE LOAD BASE MOVEMENT LBS INCHES LBS INCHES	4E13.5
6	NEGATIVE UPWARD DISPLACEMENT	
7	TOP LOAD TOP MOVEMENT BASE LOAD BASE MOVEMENT LBS INCHES LBS INCHES	4E13.5
8	STRUC LOAD(LBS) SOILP(PSF) ACTIVE DEPTH(FT) (Line 14 of Table C-2)	F10.0,2F10.2
9	BELL RESTRAINT(LBS)=	F13.2
10	(If STRUL < 0.0 See Line 14, Table C-2) FIRST ESTIMATE OF PULLOUT RESTRAINT(LBS)=	F13.2
11	LOAD-DISPLACEMENT BEHAVIOR	
12	(If LO <2) EFFECTS OF ADJACENT SOIL	
13	INITIAL BASE FORCE(LBS)= (If LO = 0) BASE FORCE(LBS)=	F13.2
14	DISPLACEMENT(INCHES)= FORCE= POUNDS	F8.4,F12.2
15	ITERATIONS=	I5
16	DEPTH(FT) LOAD(LBS) SHAFT MVMT(IN) SOIL MVMT(IN)	F7.2,2X,E13.5, 2F15.5

TABLE 3 (Continued)

c. Description of Calculations

Item	Program Prints	Description
1	BEARING CAP...	End bearing capacity, pounds
2	DOWNWARD DISPL	Load-displacment Behavior for zero soil movement in downward direction for IJ points
3	POINT BEARING	Load at bottom of shaft prior to shaft load-transfer calculation, pounds
4	DEPTH	Depth, ft
	LOAD TRANS	Load transferred at given depth along shaft, pounds
	TOTAL LOAD	Total load on shaft at given depth, pounds
	COM OF INCR	Incremental shaft compression at given depth, in.
	TOTAL MOVMT	Shaft-soil relative movement at given depth, in.
	ITER	Number of iterations to complete calculation
5	TOP LOAD	Load at top of shaft, pounds
	TOP MOVEMENT	Displacement at top of shaft, in.
	BASE LOAD	Load at bottom of shaft, pounds
	BASE MOVEMENT	Displacement at bottom of shaft, in.
6	NEGATIVE UPWARD	Load-displacement Behavior for zero soil movement in upward direction for IJ points
7	Same as item 5	
8	STRUC LOAD(LBS)	Load applied on top of shaft, pounds
	SOILP(PSF)	Pressure applied on top of adjacent soil, psf
	ACTIVE DEPTH	Depth of soil beneath ground surface subject to soil heave, ft
9	BELL RESTRAINT	Restraining resistance of bell, pounds
10	FIRST ESTIMATE	Initial calculation of pullout resistance prior to iterations for structural loads less than zero, pounds
11	LOAD-DISPLACE	Load-shaft movement distribution for given structural load
12	EFFECTS OF ADJ	Effects of soil movement considered in load-displacement behavior
13	INITIAL BASE	Initial calculation of force at bottom of shaft prior to iterations
14	DISPLACEMENT	Displacement at bottom of shaft after 100 iterations, in.

TABLE C-3 (Concluded)

Item	Program Prints	Description
	FORCE=	Force at bottom of shaft, pounds after 100 iterations, pounds
15	ITERATIONS	Total number of iterations to converge to solution
16	DEPTH(FT)	Depth, ft
	LOAD(LBS)	Load at given depth, pounds
	SHAFT MVMT(IN)	Shaft displacement, in.
	SOIL MVMT(IN)	Soil movement,in.

b. **Output Data**

Results of the computations by AXILTR are printed in LTROUT.TXT illustrated in Table C-3b. Table C-3c provides a description of calculations illustrated in Table C-3b.

(1) Load-displacement data are placed in file LDCOM.DAT for plotting by graphic software.

(2) Load-depth data for a given applied load on the pile top are placed in file LDEP.DAT for plotting by graphic software.

(3) Displacement-depth data for a given applied load on the pile top are placed in file MDEP.DAT for plotting by graphic software.

C-2. **Applications**

The pullout, uplift and downdrag capabilities of AXILTR are illustrated by two example problems. The accuracy of these solutions can be increased by using more soil layers, which increases control over soil input parameters such as swell pressure, maximum past pressure, and shear strength.

a. **Pullout and Uplift**

Table C-4 illustrates input data required to determine performance of a 2 ft diameter drilled shaft 50 ft long constructed in an expansive clay soil of two layers, NMAT = 2. The shaft is underreamed with a 5-ft diameter bell. Soil beneath the shaft is nonexpansive. The shaft is subject to a pullout force of 300 kips. Refer to Figure C-1 for a schematic representation of this problem.

TABLE C-4

Listing of Data Input for Expansive Soil, File DATLTR.TXT

```
     EXPANSIVE SOIL
    2   50  1.0  40.      2   50   16
    6    0    8     0.0       2.0       5.00
    0
  1  2.68   .8  30.    4800.    .1     .2   2000.   .0    .7   7000.
  2  2.65   .37 13.1   6000.    .1     .2   4000.   .0   2.   10000.
   0.9       0.9
    1    1
   41    2
   50    2
  .900    1.600E+05
    2
    4.333E 08   .0
    4.333E 08 50.0
-300000.        .0            50.
      0.        .0            -1.0
    0
```

(1) **Bearing Capacity**

The alpha skin friction and local shear base capacity models are selected. Option to input the reduction factor α (I = 6) was used. The selected α's for the two soils is 0.9. A high α was selected because expansive soil increases pressure against the shaft, which may raise the skin friction.

(2) **Load-Transfer Models**

The Kraft, Ray, and Kagawa skin friction and the Vijayvergiya base load-transfer models (K = 8) were selected. Two points for the elastic modulus of the shaft concrete were input into the program.

(3) **Results**

The results are plotted in Figure C-2 for a pullout force of 300,000 pounds. Results of the computation placed in file "LTROUT.TXT" are shown in Table C-5.

(a) Total and base ultimate bearing capacity is about 1,200 and 550 kips, respectively, Figure C-2a. Base and total capacity is 250 and 600 kips, respectively, if settlement is limited to 0.5 inch, which is representative of a FS of approximately 2.

(b) The distribution of load with depth, Figure C-2b, is a combination of the shapes indicated in Figures 5-9 and 5-10 because both pullout and uplift forces must be resisted.

(c) The shaft will heave approximately 0.7 inch, while the soil heaves more than 11 inches at the ground surface, Figure C-2c.



a. LOAD-DISPLACEMENT BEHAVIOR, FILE LDCOM



b. AXIAL LOAD-DEPTH BEHAVIOR, FILE LDEP

Figure C-2. Plotted output for pullout and uplift problem



c. DISPLACEMENT-DEPTH BEHAVIOR, FILE MDEP

Figure C-2. Concluded

TABLE C-5

Listing of Output, File LTROUT.TXT

```
    EXPANSIVE SOIL
NMAT=    2   NEL=   50   DX=  1.00 FT   GWL=  40.00 FT
  LO=    2   IQ (SHAFT INC)=   50   IJ (NO.LOADS)=   16

I=    6   J=    0   K=    8          SOILP=      0.00 PSF
DS=      2.00 FT
DB=      5.00 FT

LOCAL SHEAR FAILURE AT BASE - LLL=    0

MAT  GS    EO   WO(%)  PS(PSF)   CS     CC   CO(PSF)  PHI   K   PM(PSF)

 1  2.68  0.80 30.00  4800.   0.10   0.20  2000.  0.00  0.70  7000.
 2  2.65  0.37 13.10  6000.   0.10   0.20  4000.  0.00  2.00 10000.

ALPHA=   0.90000   0.90000

 ELEMENT     NO OF SOIL
    1            1
    2            1
    .            1
    .            1
   40            1
   41            2
```

TABLE C-5 (Continued)

42	2
.	2
.	2
50	2

REDUCTION FACTOR= 0.900 SHEAR MODULUS= 0.160E+06

E SHAFT(PSF) AND DEPTH(FT):
0.433E+09 0.00 0.433E+09 50.00

BEARING CAPACITY= 549778.69 POUNDS

DOWNWARD DISPLACEMENT

TOP LOAD LBS	TOP MOVEMENT INCHES	BASE LOAD LBS	BASE MOVEMENT INCHES
0.24017E+06	0.17714E+00	0.10946E+06	0.99065E-01
0.34507E+06	0.26781E+00	0.13882E+06	0.15855E+00
0.45773E+06	0.37719E+00	0.16817E+06	0.23526E+00
0.58421E+06	0.50996E+00	0.19753E+06	0.33139E+00
0.71040E+06	0.66509E+00	0.22688E+06	0.44915E+00
0.82982E+06	0.84256E+00	0.25624E+06	0.59070E+00
0.92817E+06	0.10432E+01	0.28559E+06	0.75826E+00
0.97601E+06	0.12587E+01	0.31494E+06	0.95401E+00
0.10054E+07	0.14978E+01	0.34430E+06	0.11801E+01
0.10347E+07	0.17694E+01	0.37365E+06	0.14388E+01
0.10641E+07	0.20758E+01	0.40301E+06	0.17323E+01
0.10934E+07	0.24192E+01	0.43236E+06	0.20627E+01
0.11228E+07	0.28017E+01	0.46172E+06	0.24323E+01
0.11521E+07	0.32256E+01	0.49107E+06	0.28432E+01
0.11815E+07	0.36930E+01	0.52042E+06	0.32977E+01
0.12108E+07	0.42061E+01	0.54978E+06	0.37979E+01

NEGATIVE UPWARD DISPLACEMENT

TOP LOAD LBS	TOP MOVEMENT INCHES	BASE LOAD LBS	BASE MOVEMENT INCHES
-0.18590E+05	-0.37138E-02	0.00000E+00	0.00000E+00
-0.31134E+05	-0.16708E-01	0.00000E+00	-0.10000E-01
-0.43689E+05	-0.29706E-01	0.00000E+00	-0.20000E-01
-0.68793E+05	-0.55704E-01	0.00000E+00	-0.40000E-01
-0.11899E+06	-0.10770E+00	0.00000E+00	-0.80000E-01
-0.21806E+06	-0.21160E+00	0.00000E+00	-0.16000E+00
-0.38024E+06	-0.41089E+00	0.00000E+00	-0.32000E+00
-0.61240E+06	-0.78911E+00	0.00000E+00	-0.64000E+00
-0.69610E+06	-0.14531E+01	0.00000E+00	-0.12800E+01
-0.69610E+06	-0.27331E+01	0.00000E+00	-0.25600E+01
-0.69610E+06	-0.52931E+01	0.00000E+00	-0.51200E+01

TABLE C-5 (Continued)

-0.69610E+06	-0.10413E+02	0.00000E+00	-0.10240E+02
-0.69610E+06	-0.20653E+02	0.00000E+00	-0.20480E+02
-0.69610E+06	-0.41133E+02	0.00000E+00	-0.40960E+02
-0.69610E+06	-0.82093E+02	0.00000E+00	-0.81920E+02
-0.69610E+06	-0.16401E+03	0.00000E+00	-0.16384E+03

STRUC LOAD(LBS)	SOILP(PSF)	ACTIVE DEPTH(FT)
-300000.	0.00	50.00

BELL RESTRAINT(LBS)= 449915.44

FIRST ESTIMATE OF PULLOUT RESTRAINT(LBS)= 541894.31

LOAD-DISPLACEMENT BEHAVIOR

INITIAL BASE FORCE(LBS)= -788275.25

DISPLACEMENT(INCHES)= -0.2475 FORCE= -667768.19 POUNDS

DISPLACEMENT(INCHES)= -0.4975 FORCE= -532357.44 POUNDS

DISPLACEMENT(INCHES)= -0.6525 FORCE= -449443.94 POUNDS

ITERATIONS= 262

DEPTH(FT)	LOAD(LBS)	SHAFT MVMT(IN)	SOIL MVMT(IN)
0.00	-0.32427E+06	-0.88276	-11.94514
1.00	-0.33520E+06	-0.87985	-10.67843
2.00	-0.34613E+06	-0.87685	-9.72980
3.00	-0.35706E+06	-0.87375	-8.92906
4.00	-0.36799E+06	-0.87055	-8.22575
5.00	-0.37892E+06	-0.86726	-7.59519
6.00	-0.38985E+06	-0.86387	-7.02274
7.00	-0.40078E+06	-0.86039	-6.49865
8.00	-0.41171E+06	-0.85681	-6.01600
9.00	-0.42264E+06	-0.85313	-5.56958
10.00	-0.43357E+06	-0.84936	-5.15537
11.00	-0.44450E+06	-0.84549	-4.77014
12.00	-0.45543E+06	-0.84152	-4.41124
13.00	-0.46636E+06	-0.83746	-4.07648
14.00	-0.47729E+06	-0.83330	-3.76401
15.00	-0.48822E+06	-0.82904	-3.47223
16.00	-0.49915E+06	-0.82469	-3.19976
17.00	-0.51008E+06	-0.82024	-2.94538
18.00	-0.52101E+06	-0.81570	-2.70805
19.00	-0.53194E+06	-0.81105	-2.48680
20.00	-0.54287E+06	-0.80632	-2.28080
21.00	-0.55380E+06	-0.80148	-2.08927

TABLE C-5 (Concluded)

22.00	-0.56473E+06	-0.79655	-1.91153
23.00	-0.57566E+06	-0.79153	-1.74696
24.00	-0.58613E+06	-0.78641	-1.59498
25.00	-0.59556E+06	-0.78120	-1.45506
26.00	-0.60381E+06	-0.77591	-1.32673
27.00	-0.61073E+06	-0.77056	-1.20953
28.00	-0.61621E+06	-0.76515	-1.10306
29.00	-0.62027E+06	-0.75970	-1.00692
30.00	-0.62304E+06	-0.75422	-0.92078
31.00	-0.62444E+06	-0.74872	-0.84428
32.00	-0.62465E+06	-0.74321	-0.77713
33.00	-0.62386E+06	-0.73771	-0.71902
34.00	-0.62223E+06	-0.73222	-0.66969
35.00	-0.61992E+06	-0.72674	-0.62887
36.00	-0.61710E+06	-0.72129	-0.59633
37.00	-0.61390E+06	-0.71587	-0.57183
38.00	-0.61049E+06	-0.71047	-0.55516
39.00	-0.60701E+06	-0.70510	-0.54610
40.00	-0.60360E+06	-0.69977	-0.54447
41.00	-0.59487E+06	-0.69448	-0.46514
42.00	-0.58401E+06	-0.68929	-0.39155
43.00	-0.57119E+06	-0.68420	-0.32363
44.00	-0.55675E+06	-0.67922	-0.26128
45.00	-0.54103E+06	-0.67439	-0.20443
46.00	-0.52416E+06	-0.66969	-0.15300
47.00	-0.50642E+06	-0.66515	-0.10692
48.00	-0.48799E+06	-0.66077	-0.06611
49.00	-0.46897E+06	-0.65655	-0.03049
50.00	-0.44944E+06	-0.65250	0.00000

STRUC LOAD(LBS)	SOILP(PSF)	ACTIVE DEPTH(FT)
0.	0.00	-1.00

b. **Downdrag**

Table C-6 illustrates input data required to solve for the performance of the same drilled shaft and soil described in the previous example problem, but the soil is wetter with a much lower swell pressure. Soil shear strength is assumed not to change significantly from the previous example. This shaft is subject to a 150-kip load in addition to the downdrag forces from the settling soil.

(1) **Bearing Capacity**

The alpha skin friction and local shear bearing capacity models are selected similar to the previous example. Option to input the reduction factor α (I = 6) was used. The selected α's are 0.55 and 0.3 for the surface and deeper soils, respectively.

TABLE C-6

Listing of Data Input for Settling Soil, File DATLTR.TXT

```
     SETTLING SOIL
    2   50  1.0  40.      2   50   16
    6    0    2     0.0       2.0       5.00
       0.010
    0
  1  2.68   .8  30.    1200.    .05    .1   2000.   .0    .7   4000.
  2  2.65   .37 13.1   6000.    .05    .1   4000.   .0   2.   10000.
   0.55      0.3
    1    1
   41    2
   50    2
    2
    4.333E 08   .0
    4.333E 08 50.0
 150000.         .0             50.
      0.         .0             -1.0
    0
```

(2) **Load-Transfer Models**

The Seed and Reese skin friction and Reese and Wright base load-transfer models were selected (K = 2). Two points for the elastic modulus of the shaft concrete were input into the program.

(3) **Results**

The results are plotted in Figure C-3 for a downward applied load of 150 kips. Results of the computation placed in file LTROUT.TXT are illustrated in Table C-7.

(a) Total and base ultimate bearing capacity, Figure C-3a, is about 550 and 880 kips, respectively. Base and total capacity is about 200 and 500 kips, respectively, if settlement is limited to 0.5 inch. The factor of safety is approximately 1.8 relative to total bearing capacity. The program does not add the vertical plunging failure lines to the curves in Figure C-3a, which leaves the calculated displacement load relationships nearly linear.

(b) The distribution of load with depth, Figure C-3b, is representative of downdrag indicated in Figure 5-11. The load on the shaft base is nearly 300 kips or double the applied load at the ground surface.

(c) The shaft will settle approximately 1 inch, while the soil settles about 2 inches at the ground surface, Figure C-3c. The soil is heaving near the ground surface, which counters the settlement from downdrag. Maximum settlement is about 3.5 inches at 10 ft of depth.



a. LOAD-DISPLACEMENT BEHAVIOR, FILE LDCOM



b. AXIAL LOAD-DEPTH BEHAVIOR, FILE LDEP

Figure C-3. Plotted output for downdrag problem



c. DISPLACEMENT-DEPTH BEHAVIOR, FILE MDEP

Figure C-3. Concluded

TABLE C-7

Listing of Output, File LTROUT.TXT

```
    SETTLING SOIL
NMAT=    2   NEL=   50   DX=  1.00 FT   GWL=  40.00 FT
  LO=    2   IQ (SHAFT INC)=   50   IJ (NO.LOADS)=   16

I=    6    J=    0   K=    2          SOILP=      0.00 PSF
DS=       2.00 FT
DB=       5.00 FT

E50=    0.100E-01

LOCAL SHEAR FAILURE AT BASE - LLL=    0


MAT  GS    EO   WO(%)  PS(PSF)   CS     CC   CO(PSF)  PHI    K   PM(PSF)

 1  2.68  0.80 30.00  1200.    0.05   0.10  2000.  0.00  0.70  4000.
 2  2.65  0.37 13.10  6000.    0.05   0.10  4000.  0.00  2.00 10000.


ALPHA=   0.55000   0.30000
```

TABLE C-7 (Continued)

ELEMENT	NO OF SOIL
1	1
2	1
.	1
.	1
40	1
41	2
42	2
.	2
.	2
50	2

E SHAFT(PSF) AND DEPTH(FT):
0.433E+09 0.00 0.433E+09 50.00

BEARING CAPACITY= 549778.69 POUNDS

DOWNWARD DISPLACEMENT

TOP LOAD LBS	TOP MOVEMENT INCHES	BASE LOAD LBS	BASE MOVEMENT INCHES
0.43825E+06	0.36209E+00	0.10946E+06	0.24071E+00
0.47316E+06	0.46787E+00	0.13882E+06	0.33163E+00
0.50252E+06	0.57771E+00	0.16817E+06	0.42854E+00
0.53187E+06	0.69319E+00	0.19753E+06	0.53108E+00
0.56122E+06	0.81401E+00	0.22688E+06	0.63896E+00
0.59058E+06	0.93992E+00	0.25624E+06	0.75193E+00
0.61993E+06	0.10707E+01	0.28559E+06	0.86977E+00
0.64929E+06	0.12061E+01	0.31494E+06	0.99228E+00
0.67864E+06	0.13461E+01	0.34430E+06	0.11193E+01
0.70800E+06	0.14904E+01	0.37365E+06	0.12507E+01
0.73735E+06	0.16389E+01	0.40301E+06	0.13862E+01
0.76671E+06	0.17915E+01	0.43236E+06	0.15259E+01
0.79606E+06	0.19481E+01	0.46172E+06	0.16695E+01
0.82541E+06	0.21085E+01	0.49107E+06	0.18170E+01
0.85477E+06	0.22727E+01	0.52042E+06	0.19682E+01
0.88412E+06	0.24405E+01	0.54978E+06	0.21231E+01

NEGATIVE UPWARD DISPLACEMENT

TOP LOAD LBS	TOP MOVEMENT INCHES	BASE LOAD LBS	BASE MOVEMENT INCHES
-0.19877E+05	-0.38437E-02	0.00000E+00	0.00000E+00
-0.44463E+05	-0.18937E-01	0.00000E+00	-0.10000E-01
-0.69052E+05	-0.34038E-01	0.00000E+00	-0.20000E-01
-0.11821E+06	-0.64239E-01	0.00000E+00	-0.40000E-01
-0.21272E+06	-0.12447E+00	0.00000E+00	-0.80000E-01

TABLE C-7 (Continued)

-0.31375E+06	-0.22746E+00	0.00000E+00	-0.16000E+00
-0.36937E+06	-0.40225E+00	0.00000E+00	-0.32000E+00
-0.36937E+06	-0.72225E+00	0.00000E+00	-0.64000E+00
-0.36937E+06	-0.13623E+01	0.00000E+00	-0.12800E+01
-0.36937E+06	-0.26423E+01	0.00000E+00	-0.25600E+01
-0.36937E+06	-0.52023E+01	0.00000E+00	-0.51200E+01
-0.36937E+06	-0.10322E+02	0.00000E+00	-0.10240E+02
-0.36937E+06	-0.20562E+02	0.00000E+00	-0.20480E+02
-0.36937E+06	-0.41042E+02	0.00000E+00	-0.40960E+02
-0.36937E+06	-0.82002E+02	0.00000E+00	-0.81920E+02
-0.36937E+06	-0.16392E+03	0.00000E+00	-0.16384E+03

STRUC LOAD(LBS)	SOILP(PSF)	ACTIVE DEPTH(FT)
150000.	0.00	50.00

BELL RESTRAINT(LBS)= 449915.44

LOAD-DISPLACEMENT BEHAVIOR

POINT BEARING(LBS)= 37465.96

DEPTH FT	LOAD TRANS LBS	TOTAL LOAD LBS	COM OF INCR INCHES	TOTAL MVMT INCHES	ITER
0.49500E+02	0.35018E+04	0.40968E+05	0.34571E-03	0.82732E-01	2
0.48500E+02	0.35181E+04	0.44486E+05	0.37665E-03	0.83108E-01	2
0.47500E+02	0.35358E+04	0.48022E+05	0.40775E-03	0.83516E-01	2
0.46500E+02	0.35550E+04	0.51577E+05	0.43900E-03	0.83955E-01	2
0.45500E+02	0.35756E+04	0.55152E+05	0.47043E-03	0.84425E-01	2
0.44500E+02	0.35976E+04	0.58750E+05	0.50205E-03	0.84928E-01	2
0.43500E+02	0.36210E+04	0.62371E+05	0.53386E-03	0.85461E-01	2
0.42500E+02	0.36459E+04	0.66017E+05	0.56589E-03	0.86027E-01	2
0.41500E+02	0.36722E+04	0.69689E+05	0.59815E-03	0.86625E-01	2
0.40500E+02	0.37000E+04	0.73389E+05	0.63064E-03	0.87256E-01	2
0.39500E+02	0.32524E+04	0.76641E+05	0.66129E-03	0.87917E-01	2
0.38500E+02	0.32804E+04	0.79921E+05	0.69008E-03	0.88607E-01	2
0.37500E+02	0.33096E+04	0.83231E+05	0.71913E-03	0.89327E-01	2
0.36500E+02	0.33400E+04	0.86571E+05	0.74844E-03	0.90075E-01	2
0.35500E+02	0.33717E+04	0.89943E+05	0.77802E-03	0.90853E-01	2
0.34500E+02	0.34046E+04	0.93347E+05	0.80789E-03	0.91661E-01	2
0.33500E+02	0.34387E+04	0.96786E+05	0.83805E-03	0.92499E-01	2
0.32500E+02	0.34741E+04	0.10026E+06	0.86852E-03	0.93368E-01	2
0.31500E+02	0.35107E+04	0.10377E+06	0.89931E-03	0.94267E-01	2
0.30500E+02	0.35487E+04	0.10732E+06	0.93042E-03	0.95197E-01	2
0.29500E+02	0.35879E+04	0.11091E+06	0.96188E-03	0.96159E-01	2
0.28500E+02	0.36284E+04	0.11454E+06	0.99369E-03	0.97153E-01	2
0.27500E+02	0.36703E+04	0.11821E+06	0.10259E-02	0.98179E-01	2
0.26500E+02	0.37135E+04	0.12192E+06	0.10584E-02	0.99237E-01	2
0.25500E+02	0.37581E+04	0.12568E+06	0.10913E-02	0.10033E+00	2

TABLE C-7 (Continued)

0.24500E+02	0.37857E+04	0.12946E+06	0.11246E-02	0.10145E+00	2
0.23500E+02	0.38093E+04	0.13327E+06	0.11581E-02	0.10261E+00	2
0.22500E+02	0.38337E+04	0.13711E+06	0.11918E-02	0.10380E+00	2
0.21500E+02	0.38588E+04	0.14097E+06	0.12257E-02	0.10503E+00	2
0.20500E+02	0.38845E+04	0.14485E+06	0.12598E-02	0.10629E+00	2
0.19500E+02	0.39110E+04	0.14876E+06	0.12941E-02	0.10758E+00	2
0.18500E+02	0.39382E+04	0.15270E+06	0.13287E-02	0.10891E+00	2
0.17500E+02	0.39661E+04	0.15667E+06	0.13636E-02	0.11027E+00	2
0.16500E+02	0.39947E+04	0.16066E+06	0.13987E-02	0.11167E+00	2
0.15500E+02	0.40241E+04	0.16468E+06	0.14340E-02	0.11311E+00	2
0.14500E+02	0.40542E+04	0.16874E+06	0.14696E-02	0.11458E+00	2
0.13500E+02	0.40850E+04	0.17282E+06	0.15055E-02	0.11608E+00	2
0.12500E+02	0.41166E+04	0.17694E+06	0.15417E-02	0.11762E+00	2
0.11500E+02	0.41490E+04	0.18109E+06	0.15781E-02	0.11920E+00	2
0.10500E+02	0.41821E+04	0.18527E+06	0.16148E-02	0.12082E+00	2
0.95000E+01	0.42159E+04	0.18949E+06	0.16518E-02	0.12247E+00	2
0.85000E+01	0.42506E+04	0.19374E+06	0.16891E-02	0.12416E+00	2
0.75000E+01	0.42860E+04	0.19802E+06	0.17268E-02	0.12588E+00	2
0.65000E+01	0.43222E+04	0.20235E+06	0.17647E-02	0.12765E+00	2
0.55000E+01	0.43592E+04	0.20670E+06	0.18030E-02	0.12945E+00	2
0.45000E+01	0.43970E+04	0.21110E+06	0.18416E-02	0.13129E+00	2
0.35000E+01	0.44355E+04	0.21554E+06	0.18805E-02	0.13317E+00	2
0.25000E+01	0.44749E+04	0.22001E+06	0.19198E-02	0.13509E+00	2
0.15000E+01	0.45152E+04	0.22453E+06	0.19594E-02	0.13705E+00	2
0.50000E+00	0.45562E+04	0.22908E+06	0.19994E-02	0.13905E+00	2

INITIAL BASE FORCE(LBS)= 355177.69

ITERATIONS= 81

DEPTH(FT)	LOAD(LBS)	SHAFT MVMT(IN)	SOIL MVMT(IN)
0.00	0.14992E+06	0.98875	2.15238
1.00	0.15721E+06	0.98740	2.58505
2.00	0.16451E+06	0.98598	2.85868
3.00	0.17180E+06	0.98450	3.05836
4.00	0.17909E+06	0.98295	3.20933
5.00	0.18638E+06	0.98134	3.32392
6.00	0.19367E+06	0.97967	3.40946
7.00	0.20096E+06	0.97793	3.47082
8.00	0.20825E+06	0.97612	3.51146
9.00	0.21554E+06	0.97425	3.53398
10.00	0.22283E+06	0.97232	3.54040
11.00	0.23013E+06	0.97033	3.53233
12.00	0.23742E+06	0.96827	3.51109
13.00	0.24471E+06	0.96614	3.47778
14.00	0.25200E+06	0.96395	3.43333
15.00	0.25929E+06	0.96170	3.37853

TABLE C-7 (Concluded)

16.00	0.26658E+06	0.95938	3.31409
17.00	0.27387E+06	0.95700	3.24058
18.00	0.28116E+06	0.95455	3.15857
19.00	0.28845E+06	0.95204	3.06850
20.00	0.29575E+06	0.94946	2.97082
21.00	0.30304E+06	0.94683	2.86589
22.00	0.31033E+06	0.94412	2.75408
23.00	0.31762E+06	0.94135	2.63568
24.00	0.32491E+06	0.93852	2.51098
25.00	0.33220E+06	0.93563	2.38025
26.00	0.33949E+06	0.93267	2.24373
27.00	0.34678E+06	0.92964	2.10165
28.00	0.35407E+06	0.92655	1.95420
29.00	0.36137E+06	0.92340	1.80157
30.00	0.36866E+06	0.92018	1.64396
31.00	0.37595E+06	0.91690	1.48152
32.00	0.38324E+06	0.91355	1.31441
33.00	0.39019E+06	0.91014	1.14278
34.00	0.39292E+06	0.90669	0.96503
35.00	0.38861E+06	0.90325	0.77876
36.00	0.38207E+06	0.89985	0.58423
37.00	0.37554E+06	0.89651	0.38165
38.00	0.36901E+06	0.89323	0.17124
39.00	0.36248E+06	0.89000	-0.04679
40.00	0.35595E+06	0.88684	-0.27224
41.00	0.34864E+06	0.88373	-0.23257
42.00	0.34133E+06	0.88069	-0.19578
43.00	0.33403E+06	0.87771	-0.16181
44.00	0.32672E+06	0.87480	-0.13064
45.00	0.31941E+06	0.87195	-0.10222
46.00	0.31211E+06	0.86917	-0.07650
47.00	0.30480E+06	0.86645	-0.05346
48.00	0.29749E+06	0.86380	-0.03305
49.00	0.29018E+06	0.86121	-0.01524
50.00	0.28288E+06	0.85868	0.00000

STRUC LOAD(LBS)	SOILP(PSF)	ACTIVE DEPTH(FT)
0.	0.00	-1.00

C-3. **Listing**

A Fortran listing of the computer program is provided in Table C-8 in case modifications may be required for specific applications.

TABLE C-8

Listing

```
C     PREDICTION OF VERTICAL SHAFT MOVEMENT
C     DEVELOPED BY LD JOHNSON
      PARAMETER (NL=40,NM=100)
      COMMON AREA,DB,DX,DS,DEL,PI,P(NM),IE(NM),SOILP,NEL,NBX,KOPT
      COMMON/BAS/FORCE,QBULT,CS(NL),CC(NL),EO(NL),PM(NL),AREAB,E50
      COMMON/SHA/ALGTH,AK(NL),B(NL),C(NL),DELT(NM),DELS(NM),
     1   EP(NM),TOT(NM),ZEP(NM),IOPT,JOPT,LOPT,NC,JL,
     2   NCC,FS(11),ZEPP(11),NCURVE(11),ALPHAA(NL)
      DIMENSION TL(NL),TM(NL),R(NL),S(NL),HED(20),T(11,11),Z(11,2),
     1  PS(NL)
      DATA(T(I,1),I=1,11)/.0,.3,.6,.75,.9,.95,1.,1.,1.,1.,1./,
     1   (Z(I,1),I=1,11)/.0,.05,.1,.15,.2,.23,.3,.45,.75,1.05,1.5/
    7 CONTINUE
      Z(1,2)=0.0
      JL=1
      GAW=62.43
      EP(1)=1.E35
      PI=3.1415926
    6 NNN=1
      NC=1
      OPEN(5,FILE='DATLTR.TXT')
      OPEN(6,FILE='LTROUT.TXT')
      OPEN(7,FILE='\GRAPHER\AXIC.DAT')
      OPEN(8,FILE='\GRAPHER\AXIT.DAT')
      OPEN(9,FILE='\GRAPHER\AXIL.DAT')
      READ(5,1) (HED(I),I=1,20)
      WRITE(6,1) (HED(I),I=1,20)
    1 FORMAT(20A4)
      DO 9 I=1,NM
      DELS(I)=0.0
    9 DELT(I)=0.0
      READ(5,2) NMAT,NEL,DX,GWL,LO,IQ,IJ
    2 FORMAT(2I5,2F6.2,4I5)
      WRITE(6,3)NMAT,NEL,DX,GWL,LO,IQ,IJ
    3 FORMAT(//1X,'NMAT=',I5,3X,'NEL=',I5,3X,'DX=',F6.2,' FT',3X,
     1  'GWL= ',F6.2,' FT',/,3X,'LO=',I5,3X,'IQ (SHAFT INC)=',I5,
     2  3X,'IJ (NO.LOADS)=',I5)
      LOPT=LO
      NNP=NEL+1
      READ(5,4)IOPT,JOPT,KOPT,SOILP,DS,DB
    4 FORMAT(3I5,3F10.3)
      WRITE(6,5)IOPT,JOPT,KOPT,SOILP,DS,DB
    5 FORMAT(//1X,'I=',I5,3X,'J=',I5,3X,'K=',I5,10X,'SOILP=',F10.2,
     1  ' PSF',/,1X,'DS=',F10.2,' FT',/,1X,'DB=',F10.2,' FT')
      IF(KOPT.EQ.2.OR.KOPT.EQ.5.OR.KOPT.EQ.9)READ(5,11)E50
   11 FORMAT(E13.3)
```

TABLE C-8 (Continued)

```
      IF(KOPT.EQ.2.OR.KOPT.EQ.5.OR.KOPT.EQ.9)WRITE(6,12)E50
   12 FORMAT(//1X,'E50=',E13.3)
      IU=IJ
      READ(5,13)LLL
   13 FORMAT(I5)
      IF(LLL.EQ.0)WRITE(6,14)LLL
   14 FORMAT(//1X,'LOCAL SHEAR FAILURE AT BASE - LLL=',I5)
      IF(LLL.EQ.1)WRITE(6,15)LLL
   15 FORMAT(//1X,'GENERAL SHEAR FAILURE AT BASE - LLL=',I5)
      WRITE(6,10)
   10 FORMAT(//1X,'MAT  GS   EO   WO(%)  PS(PSF)   CS    CC   CO(PSF)',
     1 '  PHI   K   PM(PSF)',/)
    8 READ(5,16)M,TL(M),EO(M),TM(M),PS(M),CS(M),CC(M),C(M),B(M),AK(M),
     1 PM(M)
   16 FORMAT(I3,3F6.2,F7.0,2F7.2,F7.0,2F6.2,F7.0)
      WRITE(6,16)M,TL(M),EO(M),TM(M),PS(M),CS(M),CC(M),C(M),B(M),AK(M),
     1 PM(M)
      B(M)=B(M)*PI/180.
      IF(PM(M).LT.PS(M)) PM(M) = PS(M)
      IF(NMAT-M)18,20,8
   18 PRINT 17,M
   17 FORMAT(17HERROR IN MATERIAL,I5)
      GO TO 8
   20 IF(IOPT.NE.6)GO TO 23
      READ(5,21)(ALPHAA(I),I=1,NMAT)
   21 FORMAT(2(7F10.5))
      WRITE(6,22)(ALPHAA(I),I=1,NMAT)
   22 FORMAT(//1X,'ALPHA=',2(7F10.5))
   23 L=0
      WRITE(6,25)
   25 FORMAT(//1X,'ELEMENT',5X,'NO OF SOIL',/)
   30 READ(5,26)M,IE(M)
   26 FORMAT(2I5)
   35 L=L+1
      IF(M-L)45,45,40
   40 IE(L)=IE(L-1)
      GO TO 35
   45 IF(NEL-L)50,50,30
   50 CONTINUE
      DO 52 I=1,NEL
      WRITE(6,53)I,IE(I)
   52 CONTINUE
   53 FORMAT(1X,I5,10X,I5)
C     EFFECTIVE STRESS IN SOIL
      P(1)=0.0
      DXX=DX
      DO 55 I=2,NNP
      MTYP=IE(I-1)
```

TABLE C-8 (Continued)

```
      WC=TM(MTYP)/100.
      GAM=TL(MTYP)*GAW*(1.+WC)/(1.+EO(MTYP))
      IF(DXX.GT.GWL)GAM=GAM-GAW
      P(I)=P(I-1)+DX*GAM
      DXX=DXX+DX
   55 CONTINUE
      IF(KOPT.LE.6)GO TO 56
      READ(5,57)RFF,GG
   57 FORMAT(F6.3,E13.3)
      WRITE(6,59)RFF,GG
   59 FORMAT(//1X,'REDUCTION FACTOR=',F6.3,3X,'SHEAR MODULUS=',E13.3)
      HALF=FLOAT(IQ)/2.
      IJK=IFIX(HALF)
      MTYP=IE(IJK)
      PH=B(MTYP)
      SHR=C(MTYP)+AK(MTYP)*SIN(PH)*P(IJK)/COS(PH)
      CONS=SHR*DS/(2.*GG)
      D1=1.5*FLOAT(IQ)*DX/DS
      DO 58 I=1,11
      CONC=(D1-T(I,1)*RFF)/(1.-T(I,1)*RFF)
      Z(I,1)=12.*T(I,1)*CONS*ALOG(CONC)
   58 CONTINUE
C     LOAD-DEFLECTION BEHAVIOR OF SHAFT
   56 IF(KOPT.LE.2.OR.KOPT.GT.6)GO TO 72
      READ(5,62)NCA
   62 FORMAT(I5)
      WRITE(6,60)NCA
   60 FORMAT(//1X,'NO. OF LOAD-TRANSFER CURVES(<12)?=',I5)
      DO 65 I=1,NCA
      WRITE(6,61)I
   61 FORMAT(/1X,'CURVE ',I5,/)
      II=I+1
      READ(5,63)(T(M,II),M=1,11)
   63 FORMAT(11F6.2)
      WRITE(6,64)(T(M,II),M=1,11)
   64 FORMAT(1X,'RATIO SHR DEV,M=1,11 ARE',/1X,11F6.3)
   65 CONTINUE
      DO 67 M=2,11
      READ(5,66)S(M)
      IF(S(M).LT..000001)GO TO 68
      JL=2
      Z(M,2)=S(M)
      WRITE(6,69)M,S(M)
   67 CONTINUE
   68 CONTINUE
   66 FORMAT(F6.3)
   69 FORMAT(/1X,'MOVEMENT(IN.) FOR LOAD TRANSFER M=',I5,' IS ',F6.3,
     1   '  INCHES')
```

TABLE C-8 (Continued)

```
   72 IF(IOPT.LE.4.OR.IOPT.EQ.6)GO TO 78
      READ(5,81)NCC
   81 FORMAT(I5)
      WRITE(6,82)NCC
   82 FORMAT(//1X,'NO OF SKIN FRICTION-DEPTH TERMS(<12)? ARE',I5,/1X,
     1  'SKIN FRICTION(PSF)  DEPTH(FT)  CURVE NO',//)
      DO 89 M=1,NCC
      READ(5,83)FS(M),ZEPP(M),NCUR
      WRITE(6,77)FS(M),ZEPP(M),NCUR
   77 FORMAT(3X,F10.3,5X,F10.3,5X,I5)
   83 FORMAT(2F10.3,I5)
      NCURVE(M)=1+NCUR
   89 CONTINUE
   78 IF(JOPT.EQ.1)GO TO 84
      READ(5,81)NC
      READ(5,92)(EP(M),ZEP(M),M=1,NC)
   92 FORMAT(E13.3,F6.2)
      WRITE(6,102)(EP(M),ZEP(M),M=1,NC)
  102 FORMAT(//1X,'E SHAFT(PSF) AND DEPTH(FT):',/1X,4(E13.3,2X,F6.2),
     1  /1X,4(E13.3,2X,F6.2),/1X,4(E13.3,2X,F6.2),/1X,4(E13.3,2X,F6.2))
   84 UU=100.
      MM=0
      ALGTH=DX*FLOAT(IQ)
      AREA=(PI/4.)*DS**2.
      AREAB=(PI/4.)*DB**2.
      NBX=IQ+1
      IF(NBX.LT.NNP)MAT=IE(NBX)
      IF(NBX.EQ.NNP)MAT=IE(NEL)
      QBU=9.*C(MAT)
      IF(LLL.LE.0.1)QBU=7.*C(MAT)
      IF(B(MAT).LT.0.001)GO TO 86
      PH=B(MAT)
      TA=SIN(PH)/COS(PH)
      XNQ=(PH/2.)+45.*PI/180.
      IF(LLL.EQ.0)GO TO 74
      TB=(1.5*PI-PH)*TA
      XNQ=EXP(TB)/(2.*(COS(XNQ))**2.)
      GO TO 76
   74 XNQ=(1.+TA)*EXP(TA)*(SIN(XNQ)/COS(XNQ))**2.
   76 XNC=(XNQ-1.)/TA
      QBU=C(MAT)*XNC+(SOILP+P(NBX))*XNQ
   86 QBULT=QBU*AREAB
      NN=1
      NO=IU
      ITER=1
      S(1)=(SOILP+P(NBX))*AREAB
      IF(KOPT.NE.6)GO TO 91
      R(1)=0.0
```

TABLE C-8 (Continued)

```
      IU1=IU-1
      WRITE(6,95)S(1),IU1
   95 FORMAT(//,1X,'BASE DISPLACEMENT(IN.),BASE LOAD(LBS) >',F10.2,
     1    '  FOR',I5,'  POINTS')
      DO 94 M=2,IU
      READ(5,103)R(M),S(M)
      WRITE(6,103)R(M),S(M)
   94 CONTINUE
  103 FORMAT(F10.5,F15.3)
   91 WRITE(6,90)QBULT
   90 FORMAT(/,10X,'BEARING CAPACITY=',F13.2,'  POUNDS')
      WRITE(6,88)
   88 FORMAT(/,1X,'DOWNWARD DISPLACEMENT')
      IF(KOPT.NE.6)DLOAD=(QBULT-S(1))/(FLOAT(IU)-1.)
   87 IF(ITER.GT.1.AND.KOPT.NE.6)S(ITER)=S(ITER-1)+DLOAD
      FORCE=S(ITER)
      IF(KOPT.NE.6)CALL BASEL
      IF(KOPT.NE.6)R(ITER)=DEL
      IF(KOPT.EQ.6)DEL=R(ITER)
   98 TOT(NBX)=FORCE
      GO TO 97
   93 WRITE(6,99)
   99 FORMAT(/,1X,'NEGATIVE UPWARD DISPLACEMENT')
   96 R(ITER)=DEL
      TOT(NBX)=0.0
      S(ITER)=0.0
   97 CALL SHAFL(T,Z)
      TL(ITER)=TOT(1)
      TM(ITER)=DELT(1)
      IF(ITER-NO)100,101,101
  100 ITER=ITER+1
      IF(MM.EQ.0)GO TO 87
      XN=FLOAT(ITER-IU-2)
      DEL=-0.01*2.**XN
      GO TO 96
  101 WRITE(6,105)
  105 FORMAT(/,49H   TOP LOAD   TOP MOVEMENT   BASE LOAD  BASE MOVE,
     1   4HMENT,/,49H     LBS          INCHES          LBS        INCHES)
      DO 110 M=NN,NO
      WRITE(6,111)TL(M),TM(M),S(M),R(M)
      IF(MM.EQ.0) WRITE(7,111)TL(M),-TM(M),S(M),-R(M)
      IF(MM.EQ.1) WRITE(8,111)TL(M),-TM(M),S(M),-R(M)
  110 CONTINUE
  111 FORMAT(4E13.5)
  112 MM=MM+1
      DEL=0.0
      IF(MM.EQ.2)GO TO 125
      ITER=IU+1
```

TABLE C-8 (Continued)

```
      NO=2*IU
      NN=IU+1
      GO TO 93
C     SHAFT MOVEMENT FOR STRUCTURAL LOAD
  125 WRITE(6,121)
  121 FORMAT(/,1X,'STRUC LOAD(LBS)   SOILP(PSF)   ACTIVE DEPTH(FT)')
      READ(5,122)STRUL,SOILP,XA
      WRITE(6,1220)STRUL,SOILP,XA
  122 FORMAT(3F15.2)
 1220 FORMAT(3X,F10.0,3X,F10.2,5X,F10.2)
      IF(XA.LT.0.0)GO TO 380
      RESTU=0.0
      DD=1000.
      DEL=0.0
      DELL=0.0
      LOPT=0
      IF(LO.EQ.3)LOPT=3
      V=0.0025
      DO 123 I=1,NNP
  123 DELS(I)=0.0
      ITER=1
      STRU=STRUL
      QBR=(7./9.)*QBU
  124 RESTR=(QBR+SOILP+P(NBX-1))*PI*(DB**2.-DS**2.)/4.
      IF(DB.LE.DS)RESTR=0.0
      IF(STRU.LT.0.0)RESTU=RESTR+(QBR+ALGTH
     1   *150.)*AREA
      WRITE(6,126)RESTR
  126 FORMAT(/,1X,'BELL RESTRAINT(LBS)=',8X,F13.2)
      IF(STRU.LT.0.0)WRITE (6,127) RESTU
  127 FORMAT(/,1X,'FIRST ESTIMATE OF PULLOUT RESTRAINT(LBS)=',F13.2)
      NN=1
      NO=IU
      WRITE(6,128)
  128 FORMAT(/,20X,26HLOAD-DISPLACEMENT BEHAVIOR,/)
      IF(STRU.GT.0.0)GO TO 135
      FORCE=STRU
      EXFOR=RESTR+FORCE
      IF(EXFOR.GT.-0.01)DEL=0.0
      IF(EXFOR.GT.-0.01)GO TO 145
      STRU=EXFOR
      NN=IU+1
      NO=2*IU
  135 DO 136 M=NN,NO
      IF(M.GT.IU)GO TO 141
      IF(TL(M)-STRU)136,137,137
  141 IF(ABS(TL(M))-ABS(STRU))136,137,137
  136 CONTINUE
```

TABLE C-8 (Continued)

```
      DIFF=TL(M)-TL(M-1)
      IF(DIFF.LT.0.01)DEL=R(M)
      IF(DIFF.LT.0.01)GO TO 138
  137 DEL=R(M-1)+(R(M)-R(M-1))*(STRU-TL(M-1))/(TL(M)-TL(M-1))
  138 IF(ABS(STRU).GT.ABS(TL(M)).AND.STRU.LT.0.0)GO TO 376
      TOT(NBX)=S(M-1)+(S(M)-S(M-1))*(STRU-TL(M-1))/(TL(M)-TL(M-1))
      CALL SHAFL(T,Z)
      DEL=DELT(NBX)
C     VERTICAL DISPLACEMENT OF ADJACENT SOIL
  145 IF(LO.LT.2)WRITE(6,146)
  146 FORMAT(//20X,'EFFECT OF ADJACENT SOIL'/)
      IF(LO.GE.2)LOPT=3
      DO 150 K=2,NNP
      I=NNP+1-K
      DEPTH=DX*FLOAT(I)-DX/2.
      IF(DEPTH.GT.XA)GO TO 150
      MTYP=IE(I)
      TEM=(SOILP+(P(I)+P(I+1))/2.)/PS(MTYP)
      TEA = (SOILP+(P(I)+P(I+1))/2.)/PM(MTYP)
      TEB = PM(MTYP)/PS(MTYP)
      TEC = DX*12./(1.+EO(MTYP))
      IF(TEA.LE.1.)TDEL=CS(MTYP)*TEC*ALOG10(TEM)
      IF(TEA.GT.1.)TDEL=TEC*(CS(MTYP)*ALOG10(TEB)+CC(MTYP)*ALOG10(TEA))
      DELS(I)=TDEL+DELS(I+1)
  150 CONTINUE
C     LOAD TRANSFER TO SHAFT FROM SOIL MOVEMENT
      ITT=1
      IF(NNN.GT.1)LOPT=1
  160 TOT(NBX)=0.0
      CALL SHAFL(T,Z)
      FORCE=TOT(NBX)-TOT(1)+STRUL
      IF(ITER.EQ.1)WRITE(6,162)FORCE
  162 FORMAT(/,1X,'INITIAL BASE FORCE(LBS)=',F13.2,/)
      IF(FORCE.LT.0.0.OR.ITER.GT.1)GO TO 300
C     DOWN FORCE, POSITIVE TIP DISPLACEMENT
      BBETA=0.0
  170 IF(KOPT.NE.6)CALL BASEL
C     SHAFT MOVEMENT FOR DOWNDRAG
      IF(KOPT.NE.6)GO TO 200
      DO 163 M=2,IU
      IF(S(M)-FORCE)163,164,164
  163 CONTINUE
  164 DEL=R(M-1)+(R(M)-R(M-1))*(FORCE-S(M-1))/(S(M)-S(M-1))
  200 TOT(NBX)=FORCE
      LOPT=LO
      CALL SHAFL(T,Z)
      BETA=TOT(1)-STRUL
      IF(LO.LT.2)WRITE(6,309)ITER,FORCE,BETA
```

TABLE C-8 (Continued)

```
      IF(UU-ABS(BETA))205,350,350
  205 IF(BETA.LT.0.0)GO TO 206
      IF(BBETA.LT.-50..AND.DD.LT.101.)GO TO 350
      IF(BBETA.LT.-50..AND.DD.GT.999.)DD=100.
      BBETA=BETA
      FORCE=TOT(NBX)-DD
      IF(FORCE.LT.0.0)ITER=1
      IF(FORCE.LT.0.0)GO TO 300
      ITER=ITER+1
      GO TO 170
  206 IF(BBETA.GT.50..AND.DD.LT.101.)GO TO 350
      IF(BBETA.GT.50..AND.DD.GT.999.)DD=100.
      BBETA=BETA
      FORCE=TOT(NBX)+DD
      ITER=ITER+1
      GO TO 170
C     UPLIFT FORCE, NEGATIVE TIP DISPLACEMENT
  300 LOPT=LO
      IF(ITER.EQ.1)DELL=DELT(NBX)
      IF(LOPT.GE.1)GO TO 307
  302 WRITE(6,305)FORCE
  305 FORMAT(/,1X,'BASE FORCE(LBS)=',3X,F13.2)
  307 EXFOR=RESTR+FORCE
      IF(EXFOR.GT.0.0)GO TO 340
      IF(ITT.EQ.100)WRITE(6,311)DELL,FORCE
      IF(ITT.EQ.100)ITT=0
      DELL=DELL-V
      DEL=DELL
      IF(DEL.LT.DELS(1))GO TO 375
      LOPT=LO
      ITT=ITT+1
      ITER=ITER+1
      GO TO 160
  309 FORMAT(25X,I5,2F15.2)
  311 FORMAT(/,1X,'DISPLACEMENT(INCHES)=',3X,F8.4,4X,'FORCE=',F12.2,
     1   '  POUNDS')
C      SHAFT MOVEMENT FOR UPLIFT
  340 TOT(NBX)=FORCE
      PRINT 311,DELL,FORCE
      DEL=DELL
      LOPT=LO
      IF(LO.EQ.1)LOPT=0
      CALL SHAFL(T,Z)
  350 FORCE=TOT(NBX)-TOT(1)+STRUL
      WRITE(6,352)ITER
  352 FORMAT(/,1X,'ITERATIONS=',I10)
      IF(FORCE.GT.QBULT)WRITE(6,353)
  353 FORMAT(/,1X,'THE SOIL BEARING CAPACITY IS EXCEEDED')
```

TABLE C-8 (Continued)

```
      WRITE(6,355)
  355 FORMAT(/,1X,'DEPTH(FT)   LOAD(LBS)     SHAFT MVMT(IN)',
     1   '  SOIL MVMT(IN)')
      DXX=0.0
      DO 360 I=1,NBX
      TOTAL=DELT(I)+DELS(I)
      IF(FORCE.GT.0.0)TOTAL=TOTAL-DELS(NBX)
      WRITE(6,370)DXX,TOT(I),TOTAL,DELS(I)
      WRITE(9,370)-DXX,TOT(I),TOTAL,DELS(I)
      DXX=DXX+DX
  360 CONTINUE
  370 FORMAT(F7.2,2X,E13.5,2F15.5)
      GO TO 379
  376 WRITE(6,377)
  377 FORMAT(/,1X,'SHAFT PULLS OUT')
      GO TO 379
  375 WRITE(6,378)
  378 FORMAT(//,1X,'SHAFT UNSTABLE')
      WRITE(6,311)DELL,FORCE
  379 NNN=NNN+1
      GO TO 125
  380 READ(5,81)NON
      IF(NON.GE.1)GO TO 7
      CLOSE(5,STATUS='KEEP')
      CLOSE(6,STATUS='KEEP')
      CLOSE(7,STATUS='KEEP')
      CLOSE(8,STATUS='KEEP')
      CLOSE(9,STATUS='KEEP')
      STOP
      END
C
C
      SUBROUTINE BASEL
      PARAMETER (NL=40,NM=100)
      COMMON AREA,DB,DX,DS,DEL,PI,P(NM),IE(NM),SOILP,NEL,NBX,KOPT
      COMMON/BAS/FORCE,QBULT,CS(NL),CC(NL),EO(NL),PM(NL),AREAB,E50
      DIMENSION Q(NM)
      DXX=DX
      NNP=NEL+1
      NA=NBX+1
      BPRES=FORCE+(DB**2.-DS**2.)*(SOILP+P(NBX))*PI/4.
      IF(DB.LT.DS)BPRES=FORCE
      BPRES=BPRES/AREAB
      IF(KOPT.EQ.1.OR.KOPT.EQ.4.OR.KOPT.EQ.8)GO TO 15
      IF(KOPT.EQ.2.OR.KOPT.EQ.5.OR.KOPT.EQ.9)GO TO 14
      Q(NBX)=BPRES
      DO 5 I=NA,NNP
      PT=1.+(DB/(2.*DXX))**2.
```

TABLE C-8 (Continued)

```
      PT=PT**1.5
      Q(I)=P(I)+SOILP+(BPRES-P(NBX)-SOILP)*(1.-1./PT)
      DXX=DXX+DX
    5 CONTINUE
      DEL=0.0
      IF(NBX.GT.NEL)GO TO 20
      DO 10 I=NBX,NEL
      MTYP=IE(I)
      QQ=(Q(I)+Q(I+1))/2.
      PPP=SOILP+(P(I)+P(I+1))/2.
      F=QQ/PPP
      CON=CC(MTYP)
      IF(QQ.LT.PM(MTYP))CON=CS(MTYP)
      IF(QQ.GT.PM(MTYP))GO TO 8
      DEL=DEL+DX*12.*CON*ALOG10(F)/(1.+EO(MTYP))
      GO TO 10
    8 AA=PM(MTYP)/PPP
      AB=QQ/PM(MTYP)
      DEL=DEL+(CS(MTYP)*ALOG10(AA)+CC(MTYP)*ALOG10(AB))*DX*12./
     1 (1.+EO(MTYP))
   10 CONTINUE
      GO TO 20
   14 C1=0.76
      C2=1.5
      C3=2.*E50
      GO TO 16
   15 C1=1.
      C2=3.
      C3=0.04
   16 DEL=(((BPRES)*AREAB)/(QBULT*C1))**C2)*C3*DB*12.
      DEL=DEL-((((P(NBX)+SOILP)*AREAB)/(QBULT*C1))**C2)*C3*DB*12.
   20 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE SHAFL(T,Z)
      PARAMETER (NL=40,NM=100)
      COMMON AREA,DB,DX,DS,DEL,PI,P(NM),IE(NM),SOILP,NEL,NBX,KOPT
      COMMON/SHA/ALGTH,AK(NL),B(NL),C(NL),DELT(NM),DELS(NM),
     1 EP(NM),TOT(NM),ZEP(NM),IOPT,JOPT,LOPT,NC,JL,
     1 NCC,FS(11),ZEPP(11),NCURVE(11),ALPHAA(NL)
      DIMENSION T(11,11),Z(11,2)
      U=0.0001
      DELC=0.0
      IF(JOPT.EQ.0)GO TO 5
      EPILE=EP(1)
    5 IF(LOPT.GE.1)GO TO 16
```

TABLE C-8 (Continued)

```
      WRITE(6,10)TOT(NBX)
   10 FORMAT(19HPOINT BEARING(LBS)=,F13.2,//)
      WRITE(6,15)
   15 FORMAT(50H     DEPTH     LOAD TRANS   TOTAL LOAD   COM OF IN,
     1  20HCR  TOTAL MVMT  ITER,/,31H      FT          LBS          ,
     2  31H LBS        INCHES       INCHES)
   16 DELT(NBX)=DEL
      DO 50 K=2,NBX
      I=NBX+1-K
      MTYP=IE(I)
      NTER=1
      AKK=K-2
      BSLMT=ALGTH-(AKK*DX)
      DEPTH=BSLMT-DX/2.
      IF(IOPT.LE.4.OR.IOPT.EQ.6)GO TO 17
      DO 18 JJ=2,NCC
      IF(ZEPP(JJ)-DEPTH)18,19,19
   18 CONTINUE
   19 JK=NCURVE(JJ)
      SKF=FS(JJ-1)+(FS(JJ)-FS(JJ-1))*(DEPTH-ZEPP(JJ-1))/(ZEPP(JJ)-ZE
     1   PP(JJ-1))
   17 IF(IOPT.LE.4.OR.IOPT.EQ.6)JK=1
      DO 20 M=2,11
      IF(Z(M,JL)-ABS(DEL))20,21,21
   20 CONTINUE
      PERCP=T(11,JK)
      GO TO 25
   21 PERCP=T(M-1,JK)+((T(M,JK)-T(M-1,JK))*(ABS(DEL)-Z(M-1,JL))/
     1  (Z(M,JL)-Z(M-1,JL)))
   25 IF(IOPT.EQ.5)SHR=SKF*PERCP
      IF(IOPT.EQ.5)GO TO 35
   26 F=BSLMT*PI/ALGTH
      PH=B(MTYP)
      TA=SIN(PH)/COS(PH)
      SHR=TA*AK(MTYP)*(SOILP+(P(I)+P(I+1))/2.)+C(MTYP)
      IF(IOPT.EQ.1)SHR=SHR*SIN(F)
      IF(IOPT.EQ.2)SHR=SHR*0.6
      IF(IOPT.EQ.3)SHR=SHR*.45
      IF(IOPT.EQ.4)SHR=SHR*.3
      IF(IOPT.EQ.6)SHR=SHR*ALPHAA(MTYP)
   35 TNSLD=SHR*PERCP
      IF(DEL.LT.0.0)TNSLD=-TNSLD
      ALDINC=TNSLD*PI*DS*DX-(P(I+1)-P(I))*AREA
      IF(JOPT.EQ.1.OR.NC.EQ.1)GO TO 42
      DO 40 M=2,NC
      IF(ZEP(M)-DEPTH)40,41,41
   40 CONTINUE
   41 EPILE=EP(M-1)+(EP(M)-EP(M-1))*(DEPTH-ZEP(M-1))/(ZEP(M)-
```

TABLE C-8 (Concluded)

```
  1  ZEP(M-1))
 42 TOT(I)=ALDINC+TOT(I+1)
    AVGLD=TOT(I+1)+ALDINC/2.
    IF(JOPT.EQ.1)GO TO 46
    AEPILE=AREA*EPILE
    DELC=AVGLD*DX*12./AEPILE
    DELT(I)=DELC+DELT(I+1)-(DELS(I)-DELS(I+1))
    DELAV=((((TOT(I)+TOT(I+1))/2.)+TOT(I+1))*DX*3./AEPILE)
  1  +DELT(I+1)
    ETA=DELAV-DEL
    IF(U-ABS(ETA))45,46,46
 45 DEL=DELAV
    NTER=NTER+1
    GO TO 17
 46 IF(LOPT.GE.1)GO TO 48
    WRITE(6,55)DEPTH,ALDINC,TOT(I),DELC,DELT(I),NTER
 48 DEL=DELT(I)
 50 CONTINUE
 55 FORMAT(5E13.5,I5)
    RETURN
    END
```